

# SPARQL UPITNI JEZIK

---

PRIMERI SPARQL UPITA

# PRIMERI UPITA NAD DBPEDIA-OM

SPARQL endpoint: <http://dbpedia.org/sparql>

SPARQL editors:

- <http://yasgui.laurensrietveld.nl/>
- <http://dbpedia.org/snorql/>

Primer 1: Prikazati imena i datume rođenja i smrti svih ljudi rođenih u Londonu u periodu 1900-1950 godine; sortirati rezultate

```
PREFIX db-ont: <http://dbpedia.org/ontology/>
PREFIX foaf: <http://xmlns.com/foaf/0.1/>
SELECT DISTINCT ?person ?name ?birth ?death
WHERE {
    ?person db-ont:birthPlace <http://dbpedia.org/resource/London> .
    ?person db-ont:birthDate ?birth .
    ?person foaf:name ?name .
    OPTIONAL { ?person db-ont:deathDate ?death . }
    FILTER ((?birth > "1900-01-01"^^xsd:date) &&
            (?birth < "1950-01-01"^^xsd:date)) .
}
ORDER BY (?name)
```

Primer 2: Prikazati imena i datume rođenja svih glumaca rođenih u Londonu nakon 1930. godine koji su još uvek živi

```
PREFIX foaf: <http://xmlns.com/foaf/0.1/>
PREFIX db-ont: <http://dbpedia.org/ontology/>
SELECT DISTINCT ?person ?name ?birth
WHERE {
    ?person db-ont:birthPlace <http://dbpedia.org/resource/London> .
    ?person db-ont:occupation <http://dbpedia.org/resource/Actor> .
    ?person foaf:name ?name .
    ?person db-ont:birthDate ?birth
    FILTER (?birth > "1930-01-01"^^xsd:date) .
    FILTER NOT EXISTS { ?person db-ont:deathDate ?death. }
}
```

Obuhvatnije rešenje (uzima u obzir i slučajeve kad je zanimanje osobe predstavljeno kao običan string):

```
PREFIX foaf: <http://xmlns.com/foaf/0.1/>
```

```
PREFIX db-ont: <http://dbpedia.org/ontology/>
```

```
PREFIX db-prop: <http://dbpedia.org/property/>
```

```
SELECT DISTINCT ?person ?name ?birth
```

```
WHERE {
```

```
  ?person db-ont:birthPlace <http://dbpedia.org/resource/London> .
```

```
  { ?person db-ont:occupation <http://dbpedia.org/resource/Actor> . }
```

```
    UNION
```

```
  { ?person db-prop:occupation ?occupation
```

```
    FILTER ((str(?occupation) = "Actor") || (str(?occupation) = "Actress")) }
```

```
  ?person foaf:name ?name .
```

```
  ?person db-ont:birthDate ?birth
```

```
  FILTER (?birth > "1930-01-01"^^xsd:date) .
```

```
  FILTER NOT EXISTS { ?person db-ont:deathDate ?death. }
```

```
}
```

Primer 3: Prikazati imena svih prestonica koje pripadaju srednje evropskoj časovnoj zoni. Prikazati i izvorna imena onih gradova za koje je taj podatak raspoloživ.

```
PREFIX db-ont: <http://dbpedia.org/ontology/>
```

```
PREFIX db-prop: <http://dbpedia.org/property/>
```

```
PREFIX foaf: <http://xmlns.com/foaf/0.1/>
```

```
SELECT DISTINCT ?name ?native
```

```
WHERE {
```

```
  ?country rdf:type db-ont:Country ; db-ont:capital ?city .
```

```
  { ?city foaf:name ?name .} UNION { ?city rdfs:label ?name . }
```

```
    FILTER ( lang(?name) = "en" )
```

```
  {?city db-ont:timeZone <http://dbpedia.org/resource/Central_European_Time> . }
```

```
    UNION
```

```
  {?city db-prop:timezone ?timezone FILTER (str(?timezone) = "CET") }
```

```
  OPTIONAL { ?city db-prop:nativeName ?native } .
```

```
}
```

# PRIMERI UPITA NAD

## LINKED MOVIE DATABASE

SPARQL endpoint:

<http://data.linkedmdb.org/sparql>

SPARQL editori:

- <http://yasgui.laurensrietveld.nl/>
- <http://www.snee.com/sparql/spreadsheetSPARQL.html>

Primer 1: Prikazati imena svih glumaca koji su igrali u filmu sa zadatim glumcem, npr. Kevin Bacon-om

```
PREFIX mov: <http://data.linkedmdb.org/resource/movie/>
SELECT DISTINCT ?actorName
WHERE {
    ?kb mov:actor_name "Kevin Bacon".
    ?movie mov:actor ?kb;
           mov:actor ?actor.
    ?actor mov:actor_name ?actorName.
    FILTER (?kb != ?actor).
}
```



## Primer 2: Prikazati imena svih glumaca koji su igrali i u filmu Martin Scorsese-a i u filmu Steven Spielberg-a

```
PREFIX mov: <http://data.linkedmdb.org/resource/movie/>
SELECT DISTINCT ?actorName
WHERE {
  ?dir1  mov:director_name "Martin Scorsese".
  ?dir2  mov:director_name "Steven Spielberg".
  ?dir1movie mov:director ?dir1;
          mov:actor ?actor.
  ?dir2movie mov:director ?dir2 ;
          mov:actor ?actor .
  ?actor  mov:actor_name ?actorName.
}
```

## Primer 3: Prikazati identifikatore i nazive svih filmskih tema definisanih u Linked Movie Database

```
PREFIX skos: <http://www.w3.org/2004/02/skos/core#>
PREFIX mov: <http://data.linkedmdb.org/resource/movie/>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
SELECT DISTINCT ?subject ?title
WHERE {
  { ?movie skos:subject ?subject . }
    UNION
  { ?movie mov:film_subject ?subject . }
  ?subject rdfs:label ?title.
}
ORDER BY ?title
```

## Primer 4: Prikazati imena svih režisera koji su snimali filmove na temu mafije; sortirati rezultate

PREFIX mov: <http://data.linkedmdb.org/resource/movie/>

PREFIX skos: <http://www.w3.org/2004/02/skos/core#>

PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>

SELECT DISTINCT ?directorName

WHERE {

{?movie skos:subject ?subject .} UNION {?movie mov:film\_subject ?subject .}

?subject rdfs:label ?subjectLbl

    FILTER (strStarts(lcase(?subjectLbl), "mafia") ).

?movie mov:director ?director.

?director mov:director\_name ?directorName .

}

ORDER BY (?directorName)

U slučaju da SPARQL engine ne podržava SPARQL 1.1, ovu liniju zameniti sa:  
FILTER (regex(?subjectLbl, "^mafia", "i")) .

PRIMERI UPITA NAD

EDUBASE DATASET (DEO DATA.GOV.UK)

SPARQL Editor:

<http://openuplabs.tso.co.uk/demos/sparqleditor>

Primer 1: Prikazati imena svih škola koje pripadaju administrativnom okrugu “City of London”; sortirati rezultate prema nazivu škola

```
PREFIX sch-ont: <http://education.data.gov.uk/def/school/>
```

```
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
```

```
SELECT ?name
```

```
WHERE {
```

```
    ?school a sch-ont:School;
```

```
        sch-ont:establishmentName ?name;
```

```
        sch-ont:districtAdministrative ?district .
```

```
    ?district rdfs:label "City of London" .
```

```
} ORDER BY ?name
```

Primer 2:

Prikazati imena i adrese svih škola koje pripadaju administrativnom okrugu "City of London"; sortirati rezultate prema nazivu škola;  
napomena: podaci o adresi nisu raspoloživi za sve škole

```
PREFIX sch-ont: <http://education.data.gov.uk/def/school/>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
SELECT ?name ?address1 ?address2 ?postcode ?town
WHERE {
    ?school a sch-ont:School;
        sch-ont:establishmentName ?name;
        sch-ont:districtAdministrative ?district .
    ?district rdfs:label "City of London" .
OPTIONAL {
    ?school sch-ont:address ?address .
    ?address sch-ont:address1 ?address1 ;
        sch-ont:address2 ?address2 ;
        sch-ont:postcode ?postcode ;
        sch-ont:town ?town . }
} ORDER BY ?name
```

### Primer 3: Prikazati naziv i datum otvaranja najstarije škole u Engleskoj

```
PREFIX sch-ont: <http://education.data.gov.uk/def/school/>
SELECT ?name ?date
WHERE {
    ?school a sch-ont:School;
            sch-ont:establishmentName ?name;
            sch-ont:openDate ?date.
} ORDER BY ASC(?date)
LIMIT 1
```



# Zadaci za vežbu

- Napisati SPARQL upit koji prikazuje nazive svih filmova u kojima su igrali dvoje vaših omiljenih glumaca (sami zadajte imena tih glumaca)
  - Koristiti Linked Movie Database SPARQL Endpoint
- Napisati SPARQL upit koji prikazuje nazive škola u Engleskoj koje su otvorene tokom 2008. godine
  - Koristiti EduBase SPARQL Endpoint

# Zadaci za vežbu

- Napisati SPARQL upit koji prikazuje nazive, i ukoliko je raspoloživ broj studenata, svih univerziteta koji su osnovani u 19. veku, u zemljama koje pripadaju engleskom govornom području
  - Koristiti DBpedia SPARQL Endpoint

```
PREFIX db-ont: <http://dbpedia.org/ontology/>
PREFIX db-prop: <http://dbpedia.org/property/>
PREFIX foaf: <http://xmlns.com/foaf/0.1/>
SELECT ?name ?students
WHERE {
  ?univ rdf:type <http://dbpedia.org/ontology/University> ;
    foaf:name ?name ;
    db-prop:established ?creationDate ;
    db-ont:country ?country .
  ?country db-ont:language
    <http://dbpedia.org/resource/English_language>
OPTIONAL { ?univ db-ont:numberOfUndergraduateStudents ?students . }
  FILTER ((?creationDate > "1800-01-01"^^xsd:date) &&
    (?creationDate < "1900-01-01"^^xsd:date)) .
}
ORDER BY (?name)
```